

INGENIERÍA TÉCNICA INDUSTRIAL - ESP. MECÁNICAFUNDAMENTOS MATEMÁTICOS DE LA INGENIERÍAPráctica nº 2: Números complejos. Polinomios. Resolución de ecuaciones.

RESUMEN: Presentamos en esta práctica cómo opera MAPLE con números complejos de forma básica. En el segundo apartado estudiamos con detenimiento cómo MAPLE trata a los polinomios. Abordamos también la resolución de ecuaciones e inecuaciones. Estudiaremos un método de resolución gráfico y un método directo limitado; otros métodos directos e iterativos serán vistos en una próxima práctica.

COMANDOS MÁS IMPORTANTES:

evalc, Re, Im, abs, argument, convert, factor, expand, simplify, coeff, solve, fsolve

* * *

- Números complejos.

La unidad imaginaria en MAPLE se representa por

$$I = \sqrt{-1}$$

Un número complejo viene dado en su forma binómica por la expresión general $z = a + I b$.

MAPLE opera con las expresiones complejas siguiendo la estructura de cuerpo de los números complejos. Esto lo ilustramos ejecutando las operaciones básicas de suma, producto y división de complejos. Para ver los resultados debemos usar el comando

evalc(expresión compleja)

que juega el papel del evalf para las reales

```
[ > evalc((a+I*b) + (c+I*d));
```

```
[ > evalc((a+I*b) * (c+I*d));
```

```
[ > evalc((a+I*b) / (c+I*d));
```

```
[ > evalc((2-3*I)^3);
```

```
[ > evalc((5-I)^(2*I));
```

Son interesantes los comandos

Re(nº complejo), **Im**(nº complejo)

que nos dan por separado la parte real y la parte imaginaria de un complejo respectivamente:

```
[ > z:=3+I*sqrt(3); w:=-2+I*Pi;
```

```
> r:=Re(z*w); i:=Im(z*w);
```

A todo complejo $z = a + I b$ le podemos asociar otro par de parámetros que lo determinan completamente. Hablamos de su módulo y su argumento


$$|z| = r = \sqrt{a^2 + b^2} \qquad \arg(z) = \theta = \arctan\left(\frac{b}{a}\right)$$

En MAPLE podemos calcularlos con las órdenes

abs (*n° complejo*), **argument**(*n° complejo*)

```
[ > abs (2+I*2) ;
  argument (2+I*3) ;
```

Estos dos parámetros permiten dar una representación equivalente del complejo $z = a+I b$ en lo que llamamos su forma polar $z = r_0$ que nos permite dar la representación gráfica por medio del vector (a,b) en el plano.



MAPLE tiene un comando que nos da la expresión polar de un complejo directamente:

convert(*n° complejo*, **polar**)

```
[ > convert (3+I, polar) ;
```

que nos da el módulo y el argumento a la vez. La orden **convert** también se usa en otro tipo de problemas más generales, pero no lo veremos por ahora.

Ejercicio: Sean $z_1 = -2 + 4I$, $z_2 = 3 - 2I$. Calcular: $z_1 + z_2$, $z_1 * z_2$, $\frac{z_1}{z_2}$, $|z_1|$, $\arg(z_2)$ y la expresión polar de $z_1^{-1} z_2$ usando **convert**.

- Polinomios

Consideramos en este apartado un grupo especial de funciones, los polinomios de coeficientes reales:

$$p(x) = a_n x^n + a_{n-1} x^{(n-1)} + \dots + a_1 x + a_0$$

```
[ > p := x -> 3*x^4 - 2*x^3 - x + 1 ;
```

Debido a su sencillez, los polinomios tienen ciertas propiedades que no tienen otras funciones más complejas. MAPLE saca partido a estas ventajas.

En los ejercicios de la práctica 1 ya vimos las sentencias

factor(*expresión*), **expand**(*expresión*)

que factorizan o desarrollan una expresión respectivamente. Es muy útil el caso en que la expresión es un polinomio

```
[ > factor (25*x^6 - 20*x^5 - 46*x^4 + 40*x^3 + 17*x^2 - 20*x + 4) ;
[ > expand ( (3*x^2 - 2)^2 * (4*x - 2)^3 ) ;
```

Cuando trabajamos con cocientes de polinomios también es muy práctica la sentencia

simplify(*expresión*)

que simplifica la expresión que le señalemos:

```
[ > simplify((48*x^2-24*x+3)/(48*x^4-24*x^3-29*x^2+16*x-2));
```

Estas tres órdenes no sólo se usan en polinomios, sino que se usan también para simplificar o desarrollar otro tipo de expresiones, aunque no suelen funcionar tan bien como con polinomios. Otra sentencia muy usada cuando trabajamos con polinomios es la sentencia

coeff(*polinomio, variable, grado*)

que nos da el coeficiente que acompaña a la variable independiente del polinomio considerado en el grado pedido:

```
[ > coeff(p(x), x, 3); #coeficiente de x^3
```

```
[ > coeff(p(x), x, 0); #término independiente
```

Una ventaja de los polinomios frente a otras funciones aparece en el cálculo de sus raíces, o sea, calcular los x tal que $p(x)=0$.

Los elementos básicos para la resolución de este tipo de ecuaciones son:

- Todo polinomio de grado n tiene n raíces en el cuerpo de los complejos.
- Las raíces complejas $a+Ib$ se presentan siempre con su conjugada $a-Ib$.
- Todo polinomio de grado impar tiene al menos una raíz real.
- Un polinomio de grado par puede presentar todas sus raíces reales, todas complejas o mezcladas.

Por esto, MAPLE tiene una serie de funciones incorporadas para tratar las raíces de polinomios. Presentamos en esta sección alguna de ellas.

Método gráfico

La representación gráfica de la expresión $p(x)$ en la variable x a resolver permite tener una idea de dónde se localizan las soluciones de la ecuación. Se puede obtener una estimación de la solución al marcar con el ratón el punto. Hay que tener en cuenta que la representación que hacemos es siempre en un intervalo finito y puede que dejemos fuera de la representación alguna solución.

Ejemplo: Hallar por el método gráfico las raíces del polinomio $p(x)=x^5-3x^4-7x^3-2x^2-x+1$. Nos piden obtener los puntos de corte de la gráfica de $p(x)$ con el eje horizontal (eje X), o sea, los puntos x en que $p(x)=0$. En primer lugar, definimos la función polinómica

```
[ > p:=x->x^5-3*x^4-7*x^3-2*x^2-x+1;
```

En segundo lugar, dibujamos la gráfica en un intervalo amplio donde sepamos que existen raíces de la ecuación:

```
[ > plot(p(x), x=-2..4);
```

A partir de la gráfica anterior volvemos a dibujar a $p(x)$ cerrando cada vez más los intervalos donde están las raíces

```
[ > plot(p(x), x=0.3..0.4);
```

```
[ > plot(x^5-3*x^4-7*x^3-2*x^2-x+1, x=0.36..0.364);
```

Ejercicio: Calcular una aproximación a la raíz negativa de $p(x)$ de la misma precisión que la anterior.

Método directo de MAPLE

El software MAPLE proporciona una utilidad para resolver ecuaciones. El comando es

solve(*ecuación, variable*)

donde *ecuación* es una expresión del tipo $p(x)=0$. Si omitimos la variable resolverá todas

las que encuentre. Si omitimos el signo "=" en la ecuación, MAPLE asigna por defecto "=0" a la expresión que introduzcamos.

Esta utilidad funciona ligeramente diferente cuando se aplica a polinomios que cuando se aplica a funciones generales. Cuando se aplica a polinomios, si el polinomio es de grado 4 o menos, nos da la expresión algebraica de las soluciones exactas; cuando el grado es mayor, dependiendo del polinomio, nos dará una expresión exacta de la solución o no.

```
[ > q:=x->3*x^4+2*x^3+5*x^2+4*x-2;  
[ > solve(q(x)=0, x);
```

A veces, cuando no le es tan fácil dar con la solución exacta, la expresión de las soluciones la da de forma implícita:

```
[ > solve(x^5 - 3*x^4 - 7*x^3 - 2*x^2 - x + 1 = 0, x);
```

Para obtener una aproximación a los valores de la solución cuando MAPLE nos la da en función de RootOf, como en el caso anterior, hay que usar el comando:

```
[ > allvalues(%);
```

Cuando **solve** no es capaz de dar una buena solución o para conseguir mayor precisión en el cálculo de las soluciones, esto es, conseguir un mayor número de decimales en la aproximación, la sentencia solve se suele combinar con la sentencia

fsolve(ecuación, variable, opción)

donde *opción* puede ser

*un intervalo del tipo **variable=a..b**, que es un intervalo donde nos hemos asegurado que hay una única raíz, usando el método gráfico por ejemplo. Esto se suele usar para calcular las soluciones reales con una mejor precisión.

***complex**, lo que nos dará una aproximación a las soluciones complejas.

```
[ > q1:=x->x^7+x^6+x^5+x^4+x^3+x^2+1;  
[ > fsolve(q1(x), x, x=-2..-1);  
[ > fsolve(q1(x), x, complex);
```

- Resolución de ecuaciones

Abordamos la resolución de ecuaciones con el formato general:

$$f(x)=0$$

En el apartado anterior hemos visto el caso en que $f(x)$ sea un polinomio. El caso general es más complicado, ya que normalmente no sabemos cuántas soluciones puede tener dicha ecuación, ni siquiera si tiene alguna. Básicamente, los comandos de resolución de estas ecuaciones son los mismos que para los polinomios, pero los métodos que usa MAPLE internamente no funcionan tan bien como con los polinomios.

MAPLE no es infalible y no resuelve todas las ecuaciones que construyamos. Se basa en el manejo de las inversas de las funciones que comparecen en la ecuación, y no siempre puede combinarlas para resolverla.

Igual que para funciones polinómicas, los métodos de resolución son principalmente dos:

- Método gráfico

La representación gráfica de la expresión $f(x)$ en la variable x a resolver permite tener una idea de dónde se localizan las soluciones de la ecuación. Se puede obtener una estimación de la solución al marcar con el ratón el punto \hat{x} que tener en cuenta que la representación

que hacemos es siempre en un intervalo finito y puede que dejemos fuera de la representación alguna solución.

Ejemplo: Resolver gráficamente la ecuación $e^x = x^2$.

Nos piden determinar los puntos de corte de las funciones $f(x) = e^x$ y $g(x) = x^2$. Gráficamente podemos ubicar las soluciones de la ecuación.

```
[ > plot({exp(x), x^2}, x=-3..3);
```

La localizamos en el intervalo $[-1, 0]$. Representamos ahora la resta de las funciones en dicho intervalo

```
[ > plot(exp(x) - x^2, x=-1..0);
```

Ejercicio: ¿Qué valor podemos asignar como solución de la ecuación? Dar una estimación de la solución con el máximo número de decimales posible.

- Método directo de MAPLE

Ejemplo: Resolver la ecuación anterior con **solve**

```
[ > solve(exp(x) = x^2); evalf(%);
```

Vemos cómo la primera expresión dada no es del todo satisfactoria al comparecer una función intrínseca (la función de Lambert) que en general desconocemos. Usando el comando **evalf** obtenemos valores de las soluciones.

Ejemplo: Resolver la ecuación $x \sin(x) = \frac{1}{2}$

```
[ > f:=x->x*sin(x);
```

```
[ > solve(f(x)=0.5);
```

No nos proporciona todas las soluciones, ni una expresión general de las mismas. Entonces una vía alternativa y muy útil es echar mano del método gráfico junto con fsolve:

```
[ > plot(f(x)-0.5, x=-10..10);
```

```
[ > fsolve(f(x)=0.5, x, x=0..1);
```

```
[ > fsolve(f(x)=0.5, x, x=2..4);
```

```
[ > fsolve(f(x)=0.5, x, x=6..7);
```

Veamos otro ejemplo más paradójico de las limitaciones del comando solve.

Ejemplo: Resolver las ecuaciones a) $e^x = \cos(x)$ b) $e^{(x^5)} = \cos(x^2)$

```
[ > solve(exp(x) = cos(x));
```

```
[ > solve(exp(x^5) = cos(x^2));
```

Además, podemos asignar una variable a la solución bien para poderla usar posteriormente, bien para comprobar que es una buena solución.

```
[ > sol:=fsolve(cos(x)=x, x);
```

```
[ > x:=sol;
```

```
[ > evalf(cos(x)-x);
```

```
[ > x:='x'; #IMPORTANTE: hay que limpiar la variable x si  
queremos que siga siendo indeterminada
```

- Inecuaciones

Los métodos de estudio de ecuaciones vistos en las secciones anteriores nos permiten abordar el

estudio de inecuaciones de la forma

$$a < |f(x)| < b.$$

La idea radica en la representación gráfica de todos los elementos de este tipo de expresiones, esto es, de $|f(x)|$ y las constantes a y b .

Ejemplo: Resolver gráficamente la inecuación $3 < |2x-5| < 8$.

```
[ > plot({abs(2*x-5), 3, 8}, x=-2..8);
```

Ejemplo: Resolver gráficamente la inecuación $|x^2-3x-7| < 7/5$

```
[ > plot({abs(x^2-3*x-7), 7/5}, x=-2..5);
```

La sentencia **solve** también se puede usar en estos casos, de dos formas.

Primera: calculando los puntos de corte que se ven en la gráfica anterior

```
[ > solve(abs(x^2-3*x-7)=7/5, x);
```

```
[ > evalf(%);
```

Y así claramente, la solución será los intervalos $(\frac{3}{2} - \frac{1\sqrt{1065}}{10}, \frac{3}{2} - \frac{1\sqrt{785}}{10})$ y $(\frac{3}{2} + \frac{1\sqrt{785}}{10}, \frac{3}{2} + \frac{1\sqrt{1065}}{10})$.

Segunda: aplicando directamente la sentencia **solve** con desigualdades:

```
[ > solve(abs(x^2-3*x-7)<7/5, x);
```

Este caso tiene el inconveniente de que sólo funciona bien cuando trabajamos con funciones sencillas y, en general, no suele ofrecer una gran precisión. El primer caso se puede aplicar a funciones más complejas.

Ejercicio: Resolver la desigualdad $|3x-5| < 3$

Cuando se tiene un sistema de inecuaciones también se puede resolver de forma exacta con el **solve**, introduciendo entre llaves las funciones a estudiar.

Ejemplo: Hallar los números que verifican simultáneamente las desigualdades: $(x-1)/(3x) < 1$ y $(|x|+2)/x < 1$.

```
[ > solve({(x-1)/(3*x)<1, (abs(x)+2)/x<1}, x);
```

Ejemplo: Resolver la inecuación $3 < |2x-5| < 8$.

```
[ > solve({abs(2*x-5)<8, abs(2*x-5)>3}, x);
```

Con lo que la solución sería la unión de intervalos abiertos $(4, 13/2)$ y $(-3/2, 1)$.